
1   1 '------------------------------------------------------------------------------------
2   2 'CALCULATION OF MARKOV EQUATIONS USING EXPLICIT OR IMPLICIT FINITE DIFFERENCE METHODS
3   3 '------------------------------------------------------------------------------------
4   8 CLS
5   10 Print
6   14 Print "  ____________________________________________________________________________"
7   16 Print
8   20 Print "  MARKOV-2 (MARKOV2.BAS).  MARKOV SOLUTION OF MULTI-STATE SYSTEM PROBABILITIES"
9   25 Print "  ORIGINAL AUTHORS: S. MASSOUD AZIZI AND PHIL RUTHERFORD"
10   27 Print "  REVISION AUTHOR: PHIL RTHERFORD"
11   30 Print "  PAPER: http://www.philrutherford.com/Conf_Papers/markov.pdf"
12   31 Print "  ORIGINAL REVISION DATE: 10-12-1989"
13   32 Print "  CURRENT REVISION DATE:  11-09-2020"
14   33 Print "  RUN DATE: ";Date$;
15   34 Print " (DD-MM-YYYY)
16   35 Print "  RUN TIME: ";TIME$
17   36 Print "  RUN WITH MMBASIC.EXE (www.mmbasic.com)"
18   37 Print
19   38 Print "  COMPUTES TIME-DEPENDENT, MULTI-STATE PROBABILITIES USING EXPLICIT OR "
20   39 Print "  IMPLICIT FINITE DIFFERENCE METHODOLOGY.  THE SAMPLE PROBLEM USES DATA FOR A "
21   40 Print "  DEISEL GENERATOR TO COMPUTE PROBABILITIES OF BEING IN SIX STATES, STANDBY, "
22   41 Print "  OPERATION, TEST, PREVENTIVE MAINTENANCE, FAILURE AND REPAIR."
23   42 Print 
24   43 Print "  IF DESIRED, USER CAN EDIT STATE TRANSITION FREQUENCIES VIA THE DATA "
25   44 Print "  STATEMENTS AT PROGRAM END. SEE PAPER FOR DERIVATION OF TRANSITION "
26   45 Print "  FREQUENCIES."
27   46 Print
28   47 Print "  THE IMPLICIT METHOD IS THE PREFERABLE OPTION. THE EXPLICIT METHOD REQUIRES "
29   48 Print "  SHORTER TIME STEPS. A TIME STEP OF 4 HRS OR LESS GIVES STABLE RESULTS THAT "
30   49 Print "  COMPARE WELL TO THE IMPLICIT METHOD. A TIME STEP OF 5 HRS OR MORE IS "
31   50 Print "  UNSTABLE."
32   53 Print 
33   54 Input "  ENTER INITIAL TIME (HRS)           ";T
34   55 Input "  ENTER FINAL TIME (HRS)             ";TMAX
35   56 Input "  ENTER TIME STEP (HRS)              ";DELT
36   57 READ IMAX           'NO. OF STATES
37   58 JMAX = IMAX
38   60 Print "  NUMBER OF STATES                   ",IMAX
39   61 Input "  ENTER IMPLICIT (I) OR EXPLICIT (E) ";OPT$
40   62 Print
41   
42   63 If OPT$ = "I" Or OPT$ = "i" Then Print "  *** IMPLICIT FINITE DIFFERENCE OPTION ***" 
43   64 If OPT$ = "E" Or OPT$ = "e" Then Print "  *** EXPLICIT FINITE DIFFERENCE OPTION ***"
44   65 Print
45   
46   68 Print "                 *** LIST OF TRANSITION FREQUENCIES (/HR) *** "
47   70 Print
48   72 Dim ST$(IMAX)
49   74 Dim LAM(IMAX,JMAX)
50   76 Dim P(IMAX)
51   78 Dim Q(IMAX)
52   80 Dim A(IMAX,JMAX+1)
53   82 Dim X(IMAX)
54   84 Dim B(IMAX)
55   86 Dim S(IMAX)
56   
57   100 For I=1 To IMAX
58   110 READ ST$(I)                   'SYSTEM STATE DESCRIPTIONS
59   120 Print TAB(I*11+2); ST$(I);
60   125 Next I
61   
62   130 For I=1 To IMAX
63   135 Print TAB(I*11+2); "---------";
64   140 Next I
65   145 Print
66   



67   150 For I=1 To IMAX
68   155 For J=1 To JMAX
69   160 READ LAM(I,J)
70   165 Next J
71   170 Next I
72   
73   175 FOR I=1 TO IMAX
74   180 Print TAB(3);ST$(I);
75   185 For J=1 To JMAX
76   190 Print TAB(J*11+2);
77   195 Print LAM(I,J);
78   200 Next J
79   210 Print
80   215 Next I
81   
82   220 Print
83   225 Print "                 *** RESULTANT SYSTEM STATE PROBABILITIES ***"
84   230 Print
85   
86   235 For I=1 To IMAX
87   240 Print TAB(I*16-5); ST$(I);
88   245 Next I
89   
90   250 For I=1 To IMAX
91   255 Print TAB(I*16-5); "---------";
92   260 Next I
93   
94   265 Print
95   270 Print "  T(HR)";
96   
97   275 For I=1 To IMAX
98   280 Print TAB(I*16-5); " Pr("I")";
99   285 Next I
100   286 Print TAB((I)*16-8);
101   287 Print "Checksum"
102   
103   290 Print
104   295 Print TAB(2);T;
105   
106   300 For I=1 To IMAX
107   305 READ P(I)                      'INITIAL STATE PROBABILITIES
108   310 Print TAB(I*16-8);
109   315 Print P(I);
110   317 CHKSUM = CHKSUM + P(I)
111   320 Next I
112   321 Print TAB((I)*16-6);
113   322 Print CHKSUM 
114   
115   330 T=T+DELT                       'CALCULATE AND PRINT TIME STEPS
116   335 Print TAB(2);T;
117   
118   340 SUM=0                 'SUM IS USED IN CONJ. WITH 810-860 TO NORMALIZE PROBABILITIES
119   345 If OPT$ = "I" Or OPT$ = "i" Then Goto 420
120   
121   350 FOR I=1 TO IMAX                 'EXPLICIT METHOD CALCULATION
122   355 SOURCE=0
123   360 SINK=0
124   365 FOR J=1 TO JMAX
125   370 SOURCE=SOURCE + LAM(J,I)*P(J)
126   375 SINK=SINK + LAM(I,J)*P(I)
127   380 NEXT J
128   385 P(I)=P(I)+(SOURCE-SINK)*DELT
129   390 NEXT I
130   395 Goto 810
131   
132   420 For I=1 To IMAX                 'IMPLICIT METHOD CALCULATION



133   430 Q(I)=O
134   435 FOR J=1 TO JMAX
135   440 Q(I)=Q(I)+LAM(I,J)              'CALCULATE LHS OF IMPLICIT EQUATION
136   445 NEXT J
137   450 FOR J=1 TO JMAX
138   455 A(I,J)= -DELT*LAM(J,I)/2
139   460 IF I=J THEN A(I,J)= 1+DELT*Q(I)/2
140   465 NEXT J
141   470 X(I)=0
142   475 FOR J=1 TO JMAX
143   480 X(I)= X(I)+DELT*LAM(J,I)*P(J)/2  'CALCULATE RHS OF IMPLICIT EQUATION
144   485 NEXT J
145   490 B(I)=(1-DELT*Q(I)/2)*P(I)+X(I)
146   495 A(I,JMAX+1)=B(I)
147   500 Next I
148   
149   550 FOR K=1 TO IMAX-1               'PROCESS OF GAUSSIAN ELIMINATION
150   560 FOR I=K+1 TO IMAX
151   565 FACTOR=A(I,K)/A(K,K)
152   570 FOR J=K TO JMAX+1
153   580 A(I,J)=A(I,J)-FACTOR*A(K,J)
154   590 NEXT J
155   600 NEXT I
156   610 Next K
157   
158   700 FOR I=IMAX TO 1 STEP -1         'BACK-SUBSTITUTION CALCULATION
159   710 S(I)=0
160   720 FOR K=IMAX TO I+1 STEP -1
161   730 S(I)=S(I)+A(I,K)*P(K)
162   735 IF K=I THEN S(I)=0
163   740 NEXT K
164   750 P(I)=(A(I,JMAX+1)-S(I))/A(I,I)
165   760 Next I
166   
167   810 For I=1 To IMAX
168   820 SUM=SUM+P(I)                    'NORMALIZATION OF P(I)
169   830 NEXT I
170   840 FOR I=1 TO IMAX
171   850 P(I)=P(I)/SUM
172   860 Next I
173   
174   890 CHKSUM = 0
175   900 For I=1 To IMAX                 'PRINT RESULTS
176   910 Print TAB(I*16-8);
177   920 Print P(I);
178   925 CHKSUM = CHKSUM + P(I)
179   930 Next I
180   932 Print TAB((I)*16-6);
181   934 Print CHKSUM 
182   
183   950 If T = TMAX Then Goto 1085
184   960 Goto 330
185   
186   1085 Print
187   1090 Print "  __________________________________________________________________________"
188   1100 End
189   
190   1500 DATA 6
191   1700 DATA " STANDBY", "OPERATION", "  TEST", "  P.M.", " FAILURE", " REPAIR"
192   1800 DATA 0.00E+1, 1.14E-5, 1.37E-3, 6.85E-4, 3.00E-4, 0.00E+1
193   1900 DATA 4.20E-2, 0.00E+1, 0.00E+1, 0.00E+1, 3.00E-3, 0.00E+1
194   2000 DATA 4.50E-1, 1.10E-5, 0.00E+1, 3.50E-2, 1.50E-2, 0.00E+1
195   2100 DATA 3.10E-2, 1.10E-6, 2.70E-2, 0.00E+1, 3.10E-3, 1.25E-3
196   2200 DATA 0.00E+1, 0.00E+1, 0.00E+1, 0.00E+1, 0.00E+1, 2.50E-1
197   2300 DATA 7.50E-2, 0.00E+1, 4.38E-2, 0.00E+1, 6.25E-3, 0.00E+1
198   2400 DATA 1,0,0,0,0,0




